UDK 528.063.3:65.011.56
510.53:528.063.3
Originalan znanstveni rad

ALGORITMI ZA ZAOKRUŽIVANJE POVRŠINA PARCELA, GRUPA I IVIČNIH KVADRATA*

Miljenko SOLARIĆ — Zagreb**

1. UVOD

Elektronička računala uvijek računaju sve veličine na veći broj znamenaka na primjer 6 ili 12 (što ovisi od toga da li računaju normalnom točnošću ili dvostrukom točnošću), a rezultate računanja zatim ispišu printerom na onoliko znamenaka koliko želi korisnik. Pri tom elektroničko računalo neposredno prije ispisa rezultata samo automatski *zaokruži poslijednju printerom napisanu znamenku i to: na manje, ako je slijedeća znamenka u intervalu od 0 do 4,9, a na više, ako je iza zaokružene znamenke brojka između 5 i 9,9.* Na primjer: 47,2 ono zaokruži na 47, 47,5 na 48 i 47,8 na 48. Taj postupak naziva se *normalnim načinom zaokruživanja brojeva,* a valjan je kad se zaokružuje jedan rezultat (vrijednost) ili više njih nezavisnih. On je ugrađen hardware-ski u elektroničko računalo, te nisu potrebne nikakve dodatne instrukcije u kompjutorskom programu za takav način zaokruživanja rezultata prije njegovog ispisa printerom.

Međutim, ako suma većeg broja pojedinačnih zaokruženih rezultata (vrijednosti) mora biti jednaka i zaokruženoj vrijednosti zbroja nezaokruženih pojedinačnih rezultata, tj. unaprijed određenoj vrijednosti »treba« na koju se izjednačava, tada taj normalni način zaokruživanja pojedinih rezultata nije uvijek dobar (zadovoljavajući) za geodetsku upotrebu. To se može ilustrirati na ovom primjeru računanja površina parcela, datom u tabeli br. 1.
Iz te tabele se očito vidi da su izjednačene vrijednosti površina parcela u stupcu (4) dobro izračunate, jer je zbroj tih vrijednosti upravo jednak površini grupe. Međutim, sumiranjem printerom ispisanih površina napisanih u stupcu (5), (koje je kompjutor sam na osnovu hardware-ski ugrađenih instrukcija zaokružio normalnim načinom, kad mu je dana instrukcija u programu da ispiše samo cijele brojeve) dobije se da je suma svih površina parcela 7655 m^2. Dakle, pojavila se progreška od čak 10 m^2 zbog toga što je elektroničko računalo »normalnim« načinom zaokružilo rezultate pri ispisu printerom. Takve pogreške geodeti ne mogu dozvoliti u svom radu, jer svi njihovi rezultati

* Ovaj rad je finacirala Samoupravna interesna zajednica za znanstveni rad SR Hrvatske SIZ III.

** Adresa autora: Prof. dr Miljenko Solarić, Geodetski fakultet Sveučilišta u Zagrebu, 41000 Zagreb, Kačićeva 26.

Tabela br. 1: Računanje površina parcela

Broj parcele (1)	Sredina iz I i II mjer. (m²) (2)	Popravka (m²) (3)	Izjednačena površina (m²) (4)	Zaokružena izjed. površina (m²) (5)
22	30,00	0,00	30,00	30
23	13,00	0,00	13,00	13
24	18,50	0,00	18,50	19
25	668,50	0,00	668,50	669
26	774,50	0,00	774,50	775
27	527,50	0,00	527,50	528
28	504,50	0,00	504,50	505
29	308,50	0,00	308,50	309
30	167,00	0,00	167,00	167
31	711,50	0,00	711,50	712
32	212,50	0,00	212,50	213
33	131,50	0,00	131,50	132
34	94,50	0,00	94,50	95
35	105,00	0,00	105,00	105
36	30,50	0,00	30,50	31
37	56,50	0,00	56,50	57
38	43,50	0,00	43,50	44
39	43,50	0,00	43,50	44
40	91,00	0,00	91,00	91
41	596,50	0,00	596,50	597
42	450,50	0,00	450,50	451
43	298,50	0,00	298,50	299
44	14,00	0,00	14,00	14
46	3,00	0,00	3,00	3
813	208,50	0,00	208,50	209
816	230,00	0,00	230,00	230
817	43,50	0,00	43,50	44
818	73,00	0,00	73,00	73
819	215,50	0,00	215,50	216
820	116,50	0,00	116,50	117
821	387,00	0,00	387,00	387
826	475,50	0,00	475,50	476
Ima:	7644,00	0,00	7644,00	7655
Površina grupe:	7644,00			
f_p:	0,00			

računanja moraju uvijek biti jednoznačni. Uostalom, zato se i izvodi izjednačenje, da se dobije samo jedan rezultat (vrijednost), koji će biti najvjerojatniji.

Kod ručnog računanja ne bi se uopće pojavio problem, jer bi geodet razdijelio popravke, tj. zaokružio ih da zbroj zaokruženih vrijednosti bude jednak onom iznosu što »treba« biti. Tako to isto čine geodeti i u ručnom računanju poligonskih vlakova (obrazca br. 19), kad se zbog kutne nesuglasice na primjer od $F_\beta = +42''$ na 5 kutova (veznih i prelomnih) porazdijeli popravka po 8″ ili 9″, ali tako da suma popravaka bude jednaka 42″. Ustvari, geodet je u tom slučaju trebao dati popravku od 8,4″ svakom kutu, ali je zbog daljeg rada tu vrijednost zaokružio na 8″, a da bi zbroj svih popravaka bio jednak 42″ morao je popravku zaokružiti na više, tj. na 9″, kod dva kuta. Kojima trima je dao popravku 8″, a kojima dvama 9″ to je on odlučio na osnovu vlastite ocjene odnosno po svojoj volji.

Elektroničko računalo ne može zaokružiti neku vrijednost na više ili manje po vlastitoj ocjeni (volji), već samo strogo poštujući *normalni način zaokruživanja, koji nije valjan (dobar) kad suma pojedinačnih zaokruženih rezultata (vrijednosti) mora biti jednaka zaokruženoj vrijednosti sume nezaokruženih rezultata,* tj. unaprijed određenom iznosu »treba«. Zato treba u programu predvidjeti takve instrukcije da on može zaokružiti rezultate na više ili na manje prema nekom algoritmu,[2] koji najbolje odgovara konkretnom zadatku. U našim geodetskim radnim organizacijama udruženog rada neki kompjutorski programi koji se koriste ne predviđaju zaokruživanje rezultata. To bi oni trebali uzeti u obzir i pored toga što se radi obično o manjim iznosima, ako se želi osigurati jednoznačnost svih rezultata. Upravo iz tih razloga želi se u ovom članku opisati nekoliko algoritama za zaokruživanje rezultata i proanalizirati koji bi najbolje odgovarali za upotrebu u računanju površina parcela, grupa i ivičnih kvadrata.

2. ALGORITMI ZA ZAOKRUŽIVANJE IZRAČUNATIH REZULTATA

Kao što je već napisano ako se želi da je *suma pojedinačnih zaokruženih rezultata jednaka zaokruženoj vrijednosti zbroja nezaokruženih rezultata,* tada se mora u kompjutorskom programu predvidjeti takvo zaokruživanje rezultata po nekom od algoritama. U daljem tekstu, zbog kratkoće pisanja, će se pod pojmom zaokruživanja isključivo misliti na takvu vrstu zaokruživanja pojedinih rezultata kad uvjet sume mora biti zadovoljen. Ovdje će se ukratko opisati razne varijante tih algoritama i priložiti njihove potprograme napisane u programskom jeziku BASIC. Oni su prilagođeni za stolno elektroničko računalo proizvodnje Hewlett-Packard 9845 A, koje posjeduje Geodetski fakultet Sveučilišta u Zagrebu, ali tako da se mogu koristiti i na elektroničkim računalima HP 9830. Zaokruživanje je učinjeno s naredbom INT, a kako ona uvijek zaokružuje vrijednosti na manje,[3] vrijednosti L je dodano 0,5 da bi to zaokruživanje bilo jednako kao što to kompjutor radi normalnim načinom. Za elektroničko računalo HP 9845 mogao se je napisati program jednostavnije, jer ima instrukcije, koje su efikasnije od onih za stolno računalo HP 9830.

---

[2] Pod algoritmom se podrazumijeva postupak za riješavanje nekog matematičkog zadatka, a može biti prikazan matematičkim formulama i riječima ili pomoću dijagrama toka (Štefanini 1976). Drugim riječima značenje pojma algoritma može se objasniti ovako: Algoritam je opis čitavog niza operacija i pravila, tj. svih programskih koraka u nizu instrukcija, koje će od ulaznih podataka dovesti do rezultata (Dimitrijević 1981.).

[3] Naredba INT zaokružuje vrijednosti na primjer ovako: 234,8 zaokruži na 234, 234,4 na 234, —234,4 na —235 i —234,8 na —235.

a) *I varijanta*

Ukratko riječima ova prva varijanta algoritma kompjutorskog programa za zaokruživanje rezultata, kad uvjet sume mora biti zadovoljen može se izraziti ovako: *Pravoj izjednačenoj vrijednosti (L_i) dodaje se razlika (R_{i-1}), tj. ostatak od zaokruživanja prethodne pojedinačne vrijednosti ($L_{i-1,p}$), a nakon toga se ona zaokruži i tako dalje se ponavlja za slijedeće po redu izjednačene vrijednosti.* Pri tome je razlika (R_{i-1}) jednaka prethodnoj pojedinačnoj vrijednosti ($L_{i-1,p}$) minus ta ista zaokružena vrijednost ($L_{i-1,z}$). Kod prve izjednačene vrijednosti, tj. kad je i = 1 razlika $R_{1-1} = 0$. To se može izraziti u matematičkom obliku ovim jednadžbama:

$$L_{i,p} = L_i + R_{i-1} \qquad (\text{za } i = 1, 2, \ldots n) \tag{1}$$

gdje je

$$R_{i-1} = L_{i-1,p} - L_{i-1,z} \qquad (\text{za } i = 2, 3, \ldots n) \tag{2}$$

$$R_{i-1} = 0 \qquad (\text{za } i = 1)$$

Na koncu će biti zadovoljeno da je zaokružena vrijednost sume izjednačenih nezaokruženih pojedinačnih rezultata (S) jednaka zbroju zaokruženih izjednačenih pojedinačnih rezultata (Z).

Da bi se rezultate moglo zaokružiti na proizvoljan broj znamenaka na početku se pojedinačni rezultati L(I) množe s 10^{-E}, a na koncu sa 10^{E}, tj. vrati na normalni broj cijelih mjesta. Budući da je algoritam vrlo jednostavan ne prilaže se njegov dijagram toka, već samo potprogram Zaokr1.

```
10   SUB Zaokrl
20   ! Zaokruživanje pozitivnih pojedinačnih rezultata uz uvijet da
30   ! je njihov zbroj jednak zaokruženoj vrijednosti sume pojedinih
40   ! nezaokruženih rezultata.
50   ! Razlika (R), tj. ostatak koji je ostao od prethodne zaokruže-
60   ! ne vrijednosti dodaje se slijedećoj po redu vrijednosti i na-
70   ! kon toga se ona zaokruži i tako redom ponavlja.
80   !
90   ! N = ukupan broj zaokruženih pojedinih rezultata
100  ! E = broj na koliko mjesta se želi zaokružiti rezultate.
110  !     Za decimalna mjesta E je negativan, a za cijela pozitivan.
120  !     Tako će biti: za ispis na dva decimalna mjesta (npr. 244.23)
130  !     E=-2, za zaokruživanje na sva cijela mjesta (npr. 244) E=0,
140  !     a za zaokruživanje na stotine (npr. 200) E=+2.
150  ! L(I) = vrijednost od i-tog rezultata.
160  ! R = razlika, tj. ostatak koji je ostao od zaokružene vrijedno-
170  !     sti, a dodaje se slijedećoj po redu veličini L(I+1).
180  !
181  COM L(100),INTEGER N,E
190  R=0
200  FOR I=1 TO N
210  L(I)=L(I)*10^(-E)+R    ! Množi s 10**(-E), da može zaokružiti s INT
                            na proizvoljan broj mjesta
220  R=L(I)-INT(L(I)+.5)    ! razlika, tj. ostatak, koji će pribrojiti
                            slijedećem L, tj. L(I+1)
230  L(I)=L(I)*10^E         ! Množi s 10**E da vrati na stari broj znamenaka
240  NEXT I
250  SUBEND
```

Potprogram br. 1.: Zaokrl, koji računa po algoritmu I varijante

b) *II varijanta*

U ovoj varijanti algoritam zaokružuje rezultate na ovaj način: *Najprije se izračuna diferencija (D) od zaokružene vrijednosti sume izjednačenih pojedinačnih (nezaokruženih) rezultata (S) minus zbroj zaokruženih izjednačenih pojedinačnih rezultata (Z). Zatim se odrede razlike (R) od prave izjednačene pojedinačne vrijednsti (L_p) manje zaokružena ta ista vrijednost (L_z) za sve pojedine vrijednosti (L).*
Poslije toga se pronađu izjednačene pojedinačne vrijednosti (L_p), koje imaju ekstremno veliku razliku (R) s jednakim predznakom kao i diferencija (D), te ih se zaokruži na više ili na manje što ovisi od predznaka diferencije (D).
Pomoću jednadžbi to se može izraziti i ovako, da se diferencija

$$D = S - Z \tag{3}$$

razdijeli (zaokruživanjem na više ili na manje) na izjednačene pojedinačne rezultate kod kojih je razlika

$$R = L_p - L_z \tag{4}$$

ekstremno velika po veličini i s jednakim predznakom kao i diferencija (D).

Ako ima više jednakih ekstremno velikih vrijednosti (R), koje imaju jednak predznak kao i diferencija (D), zaokružuje se na više ili na manje prva izjednačena pojedinačna vrijednost (L_p), (tj. ona s manjim rednim brojem u popisu), koja ima tu ekstremno veliku razliku (R) s jednakim predznakom kao i diferencija (D). Da bi se rezultate moglo zaokružiti na proizvoljan broj znamenaka na početku se pojedinačni rezultati L(I) množe s 10^{-E}, a na koncu sa 10^{E}, tj. vrati na normalan broj cijelih mjesta.

Djagram toka algoritma II varijante je dat na slici br. 1, a u članku je priložen i potprogram Zaokr2.

c) *III varijanta*

Algoritam ove varijante *radi na istom principu kao i u II varijanti samo:* ako se utvrdi da ima više jednakih ekstremno velikih razlika (R) s istim predznakom kao i diferencija (D) *zaokruži se na više ili manje izjednačena pojedinačna vrijednost (L_p), koja ima najveću vrijednost (L_p) od svih njih s ekstremno velikom razlikom (R) jednakog predznaka kao diferencija (D).* Dijagram toka algoritma III varijante zaokruživanja rezultata je nešto složeniji, ali ipak vrlo sličan dijagramu toka algoritma II varijante, te se ga ne prilaže u članku već samo potprogram Zaokr3.

d) *IV varijanta*

U ovoj varijanti algoritam za zaokruživaje radi slično kao i u II varijanti, tj. najprije se odredi diferencija (D), kao i razlike (R), a zatim: *Diferencija (D) se razdijeli zaokruživanjem na više ili na manje onih izjednačenih rezultata (L_p) kod kojih je umnožak*

$$U = L_p \cdot R \tag{5}$$

ekstremno velik i čiji je predznak jednak kao i od diferencije (D).



Slika br. 1: dijagram toka potprograma za zaokruživanje rezultata po algoritmu II varijante.

```
10   SUB Zaokr2
20   ! Zaokruživanje pozitivnih pojedinačnih rezultata uz uvijet da
30   ! je njihov zbroj jednak zaokruženoj vrijednosti sume pojedinih
40   ! nezaokruženih rezultata.
50   ! Zaokružuje se ona vrijednost L(I),koja ima najveću razliku (R) s
60   ! predznakom od diferencije (D). Ako ima više takovih vrijednosti
90   ! zaokruži onu koja je prva po rednom broju u popisu.
91   !
100  ! N = ukupan broj zaokruženih pojedinih rezultata
110  ! E = broj na koliko mjesta se želi zaokružiti rezultate.
120  !     Za decimalna mjesta E je negativan, a za cijela pozitivan.
130  !     Tako će biti: za ispis na dva decimalna mjesta (npr. 244.23)
140  !     E=-2, za zaokruživanje na sva cijela mjesta (npr. 244) E=0,
150  !     a za zaokruživanje na stotine (npr. 200) E=+2.
160  ! L(I) = vrijednost od i-tog rezultata.
170  ! D = diferencija između zaokružene vrijednosti sume izjednače-
180  !     nih (nezaokruženih) rezultata (S) i zbroja zaokruženih izje-
190  !     dnačenih pojedinačnih rezultata (Z).
200  ! R = razlika od prave izjednačene vrijednosti L(I) minus zaokru-
210  !     žena ta ista vrijednost.
220  ! M = maksimalna vrijednost od razlike (R).
240  !
251  COM L(100),INTEGER N,E
260  Z=S=0
270  FOR I=1 TO N
280  L(I)=L(I)*10^(-E)  ! Množi s 10**(-E) da može zaokružiti s INT
                          na proizvoljan broj znamenaka.
290  Z=Z+INT(L(I)+.5)   ! Zbraja zaokružene vrijednosti L.
300  S=S+L(I)           ! Sumira nezaokružene vrijednosti L.
310  NEXT I
320  D=S-Z              ! Diferencija.
330  D1=INT(D+.5)
340  IF D1=0 THEN 490
350  FOR I=1 TO ABS(D1)
360  M=0
370  FOR J=1 TO N
380  R=(L(J)-INT(L(J)+.5))*SGN(D)   ! Razlika.
390  IF M>=R THEN 460
430  K=J
440  M=R
460  NEXT J
470  L(K)=L(K)+.5*SGN(D)   ! Vrijednosti L, koja ima ekstreman R s
                   predznakom diferencije D doda ili oduzme 0.5.
480  NEXT I
490  FOR I=1 TO N
500  L(I)=L(I)*10^E     ! Množi s 10**E, vrati rezultate od L(I) na staru
                          veličinu.
510  NEXT I
520  SUBEND
```

Potprogram br 2: Zaokr2, koji zaokružuje rezultate po algoritmu II varijante

Ako ima više jednakih umnožaka (U) s jednakim predznakom kao diferencija (D), popravka se daje prvoj vrijednosti po rednom broju u popisu. U ovoj varijanti vjerojatno će biti znatno manji broj takvih slučaja da su ekstremno veliki umnošci (U) jednaki na više mjesta, kao što se to moglo dogoditi s ostatkom (R) u II varijanti.

U članku je priložen potprogram Zaokr4, koji radi po algoritmu IV varijante.

```
10  SUB Zaokr3
20 ! Zaokruživanje po III varijanti
250 !
251 COM L(100),INTEGER N,E
260 Z=S=0
270 FOR I=1 TO N
280 L(I)=L(I)*10^(-E)
290 Z=Z+INT(L(I)+.5)
300 S=S+L(I)
310 NEXT I
320 D=S-Z
330 D1=INT(D+.5)
340 IF D1=0 THEN 490
350 FOR I=1 TO ABS(D1)
360 M=M1=0
370 FOR J=1 TO N
380 R=(L(J)-INT(L(J)+.5))*SGN(D)
390 IF M>R THEN 460
400 IF M=R THEN 420
410 M1=L(J)
420 IF M1>L(J) THEN 460
430 K=J
440 M=R
450 M1=L(J)
460 NEXT J
470 L(K)=L(K)+.5*SGN(D)
480 NEXT I
490 FOR I=1 TO N
500 L(I)=L(I)*10^E
510 NEXT I
520 SUBEND
```

Potprogram br. 3: Zaokr3, koji zaokružuje rezultate po algoritmu III varijante

```
10  SUB Zaokr4
20  ! Zaokruživanje po IV varijanti
250 !
251 COM L(100),INTEGER N,E
260 Z=S=0
270 FOR I=1 TO N
280 L(I)=L(I)*10^(-E)
290 Z=Z+INT(L(I)+.5)
300 S=S+L(I)
310 NEXT I
320 D=S-Z
330 D1=INT(D+.5)
340 IF D1=0 THEN 490
350 FOR I=1 TO ABS(D1)
360 M= 0
370 FOR J=1 TO N
380 R=(L(J)-INT(L(J)+.5))*SGN(D)
381 U=L(J)*R
390 IF M>=U THEN 460
430 K=J
440 M=U
460 NEXT J
470 L(K)=L(K)+.5*SGN(D)
480 NEXT I
490 FOR I=1 TO N
500 L(I)=L(I)*10^E
510 NEXT I
520 SUBEND
```

Potprogram br. 4: Zaokr4, koji zaokružuje rezultate po algoritmu IV varijante

3. ANALIZA POSTIGNUTE TOČNOSTI ZAOKRUŽIVANJA REZULTATA S RAZNIM VARIJANTAMA ALGORITMA I ZAKLJUČCI

Analiza postignute točnosti s pojedinim varijantama algoritama za zaokruživanje rezultata, kad mora biti zadovoljen uvjet da je zbroj pojedinih zaokruženih rezultata jednak zaokruženoj vrijednosti sume izjednačenih nezaokruženih pojedinačnih rezultata, tj. iznosu »treba«, može se učiniti detaljnim promatranjem nekih konkretnih primjera. Da bi se postigla potrebna preglednost najbolje je rezultate zaokruživanja po raznim varijantama algoritama svrstati u tabelu i tada usporediti rezultate, tj. komparirati točnost zaokruživanja. U tu svrhu se i prilaže tabela br. 2 za računanje površina parcela, a u kojoj su upisani i rezultati zaokruživanja vrijednosti računanjem po raznim načinima (algoritmima).

U stupcu (6) tabele br. 2 unešene su površine parcela zaokružene normalnim načinom, koji je hardware-ski ugrađen u kompjutor. Vidi se da nije dobijena ispravna vrijednost sume nakon zaokruživanja. Umjesto površine grupe od 75 m² dobilo se da je zbroj zaokruženih površina parcela u grupi 74 m², što je uokvireno u tabeli, jer se željelo to istaknuti. Dakle, ako se želi osigurati da je uvjet sume zadovoljen mora se koristiti kompjutorske programe, koji su napisani po nekom od algoritama za zaokruživanje rezultata. Taj zaključak je već konstatiran u uvodu, a ovdje ga se je ponovilo zbog preglednosti svih načina zaokruživanja.

Iz stupca (7) tabele br. 2 vidi se da program učinjen po algoritmu I varijante može neke vrijednosti zaokružiti na više ili manje, kako to geodet ne bi nikad učinio pri ručnom radu. Na prijmer površinu parcele br. 17 od 10,23 m² kompjutor po lagoritmu I varijante zaokruži tu površinu na 11 m². Ta je vrijednost uokvirena u tabeli br. 2, da bi se istakla upravo ona. Geodet bi prije

Tabela br. 2: Računanje površina parcela i zaokruživanje rezultata na razne načine, tj. s raznim algoritmima

Redni broj (1)	Broj parcele (2)	Sredina iz I i II mj. m² (3)	Popravka m² (4)	Izjed. površ. m² (5)	Zaokružena izjed. površina				
					norm. način (6)	I var. (7)	II var. (8)	III var. (9)	IV var. (10)
1	16	5,50	—0,14	5,36	5	5	$\boxed{6}$	$\boxed{6}$	5
2	17	10,50	—0,27	10,23	10	$\boxed{11}$	10	10	10
3	32	15,50	—0,40	15,10	15	15	15	15	15
4	33	20,50	—0,53	19,97	20	20	20	20	20
5	34	25,00	—0,65	24,35	24	24	24	24	25
	Ima:	77,00	—1,99	75,01	$\boxed{74}$	75	75	75	75
	Površina grupe:	75,00							
	f_p:	—2,00							

zaokružio na primjer površinu 5,36 m² na 6 m² ili možda i površinu 20,35 m² na 21 m². Ova mala netočnost proizlazi iz toga što se ostatak od prethodnog zaokruživanja prenosi na slijedeću veličinu po redu pa se može dogoditi pogreška u zaokruživanju rezultata od skoro 1 jedinice vrijednosti posljednje znamenke. Ova varijanta zaokruživanja rezultata je najjednostavnija, ali se mogu pojaviti, kao što se je i vidjelo, pogreške zaokruživanja, te ju se ne bi moglo preporučiti za upotrebu u računanju površina parcela i pored toga što je zadovoljen uvjet da je suma jednaka onoj što »treba« biti.

Pogleda li se stupac (8) u tabeli br. 2 vidi se da je površina parcela br. 16 zaokružena po algoritmu II varijante na 6 m² iako je njena prava izjednačena vrijednost 5,36 m². Kod tako male površine, zaokruživanjem na više, čini se velika relativna pogreška zaokruživanja. Zato bi relativna pogreška zaokruživanja bila manja, ako bi se zaokružila površina parcele br. 34 sa 24,35 m² na 25 m². Algoritam za zaokruživanje rezultata od II varijante je svakako bolji od algoritma I varijante, ali ima još neke nedostatke koje valja izbjegavati u slučajevima kad su veće razlike u veličini zaokruženih vrijednosti površina.

U stupcu (9) tabele br. 2 unešeni su zaokruženi rezultati računanja površina parcela po algoritmu treće varijante. Odmah se uviđa da su zaokružene vrijednosti po algoritmu II varijante i III varijante, tj. u stupcima (8) i (9) jednake. To se je dogodilo u ovom primjeru, zbog toga što je površina parcele br. 16 5,36 m², a parcele br. 34 24,35 m², te je ostatak (razlika) ($R = L_p - L_z$) kod normalnog načina zaokruživanja tih vrijednosti za parcelu br. 16 $R_{16} = 0{,}36$ m², a za parcelu br. 34 $R_{34} = 0,35\ m^2$. Kako je R_{16} veće od R_{34} to je zaokružena na više površina parcele br. 16. Općenito govoreći algoritam III varijante zaokruživanja rezultata dat će bolje rezultate od algoritma I i II varijante, ali i ona ima nedostatak, jer u nekim slučajevima relativna pogreška zaokruživanja površina može biti veća.

Rezultati, nakon zaokruživanja izjednačenih površina po IV varijanti, upisani su u stupcu (10) tablice br. 2. Za njih bi se moglo reći da su u skladu s onim principima zaokruživanja rezultata na više ili na manje, koje se u geodetskim računanjima nastoji obično provesti u praksi pri ručnom računanju površina. Iz toga se može zaključiti: *da je u računanjima površina parcela, grupa i ivičnih kvadrata najbolje rezultate zaokružiti na kompjutorima s programima, koji su napisani po algoritmu IV varijante.*

Na koji je način najbolje zaokružiti rezultate potrebno je promotriti u svakoj vrsti geodetskih računanja posebno.

LITERATURA:

[1] Brukner, M. (1976): Elektronička računala i programiranje, Sveučilište u Zagrebu — Geodetski fakultet, Zagreb 1976.

[2] Dimitrijević, D. (1980): Računarsko programiranje, Naučna knjiga, Beograd, 1966.

[3] Stefanini, B. (1976): Fortran V osnovni tečaj, Školska knjiga, Zagreb,1976.

[4] Žokalj, M. (1966): Fortran programiranje (Po D. D. McCreckenu), Beograd, 1966.

[5]) . . . (1977): Hewlett-Packard System 45 Desktop computer — Opereting and Programiming, 1977.

SAŽETAK

U članku se ukazuje na problem zaokruživanja rezultata u geodetskim računanjima na elektroničkim računalima kad suma zaokruženih rezultata (vrijednosti) mora biti jednaka zaokruženoj vrijednosti sume nezaokruženih rezultata, tj. unaprijed određenom iznosu »treba«. Navedene su četiri varijante zaokruživanja pojedinačnih rezultata, kad je uvjet sume zadovoljen, a zatim su analizirane njihove prednosti i nedostaci u računanju površina parcela, grupa i ivičnih kvadrata. Kao najpogodnija pokazala se četvrta varijanta algoritma. U njoj se najprije odredi diferencija (D), kao razlika između sume nezaokruženih pojedinačnih rezultata (S) i zbroja zaokruženih pojedinačnih rezultata (Z). Zatim se nađu razlike (R) između pojedinačnih nezaokruženih rezultata (L_p) i tog istog pojedinačnog zaokruženog rezultata (L_z). Poslije toga se diferencija (D) razdijeli zaokruživanjem na više ili na manje onih izjednačenih pojedinačnih rezultata (L_p) kod kojih je umnožak $U = L_p \cdot R$ ekstremno velik i čiji je predznak jednak kao i od diferencije (D). Osim toga u članku su priloženi i odgovarajući potprogrami napisani u programskom jeziku BASIC za sve četiri varijante zaokruživanja rezultata.

ABSTRACT

In this paper the problem of round writing results in geodetic calculation on computer when the sum of round writing results need to be equal to the sum of unround writing results is pointed out. The four variants of algorithm for round writing results are here described but only when the condition of sum total is fulfilled. After that their advantages and lacks in calculation area of parcel, group and sheet of map have been anlysed. The most suitable is the fourth variant of algorithm. In this variant of alogrithm as the first the difference (D) is determined between sum total of unround writing results (S) and total amount of round writing results (Z). Then the difference between unround writing single results and these same round writing single results are determined. Later on the difference is distributed by round writing to those more or less single results which have the product of multiplication $U = L_p. \cdot R$ extremly big with the same sign as the difference (D). Besides that subprograms in BASIC for all four variant of round writing results are enclosed in this paper.